Lesson Notes U40 –Application Linked to a Database
Resources: 

1. SAMS Publishing. Teach yourself Visual Basic 2005. Hour 21 
2. Graeme Summers 2003 Third Ed. Programming with Visual Basic. Introduction to VB.Net, Page 176.
In Unit 4 Outcome 1, part A, you will be required to build be an application that interacts with a database, and write a software report. In part B, you will be required to produce User Documentation (include in the report how you did it and references) as part of this application.

Visual Basic uses the ActiveX Data Object (ADO) technology, which is a set of library routines referenced at design time (but should be there by default). It also makes use of Object Linking and Embedding the database (OLEdB). These acronyms appear as software objects in the code, from which properties and methods are accessed.
This week’s Lesson notes will be a set of instructions leading you to construct two applications that each links to a database. One is basic and the other builds on your skills.

Project Instructions:

1. Read SAMS Hour 21 about a simple application that connects to a database from VB.Net allowing the user to manipulate database data from the external application interface. 

2. From the supplied diagrams below, create this database in Access and the interface using VB.Net. Then add the supplied code behind the interface elements (see below). Run and debug. It may take you about 1-2 periods complete this exercise.
3. Then continue on to create the ATM machine application based on the Summers reference. This is more challenging and may take 2-3 periods to do. But these skills will enable you to easily tackle the approaching Outcome.
4. Answer the Questions/Tasks section at the end of this document. You will be asked to make extensions to the functionality of the Summers application.

SQL Language:

Note there is a statement Select * From Contacts appearing in frmMain_Load. This is called Structured Query Language (SQL), which operates at the database level and can be used to filter out data from a table or query. In this case the statement asks for all records from the Contacts table.

SAMS VB.Net application linked to a database

Database:

The database is easily created in Access. It consists of only one table with two columns. The fields are Contact Name and State.

User Interface:

The VB.Net interface is simple. It consists only of text boxes and buttons as shown.

Below is the VB.Net code to add behind the interface elements. Hint! Copy and paste the code located inside the subroutines, not the subroutine “shell”.

VB.Net Code:
Imports System.Data.OleDb

Public Class frmMain

    ' declare variables at the module level. These are "seen" in all

    ' the subroutines of this class

    Private m_cnADONetConnection As New OleDb.OleDbConnection()

    Private m_daDataAdapter As OleDb.OleDbDataAdapter

    Private m_cbCommandBuilder As OleDb.OleDbCommandBuilder

    Private m_dtContacts As New DataTable

    Private m_rowPosition As Integer = 0

    Private Sub frmMain_FormClosed(ByVal sender As Object, ByVal e As System.Windows.Forms.FormClosedEventArgs) Handles Me.FormClosed

        m_cnADONetConnection.Close()

        m_cnADONetConnection.Dispose()

    End Sub

    Private Sub ShowCurrentRecord()

        If m_dtContacts.Rows.Count = 0 Then

            txtContactName.Text = ""

            txtState.Text = ""

            Exit Sub

        End If

        txtContactName.Text = _

           m_dtContacts.Rows(m_rowPosition)("ContactName").ToString()

        txtState.Text = _

           m_dtContacts.Rows(m_rowPosition)("State").ToString()

    End Sub

    Private Sub frmMain_Load(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MyBase.Load

        m_cnADONetConnection.ConnectionString = _

              "Provider=Microsoft.Jet.OLEDB.4.0;Data Source=C:\temp\contacts.mdb"

        m_cnADONetConnection.Open()

        m_daDataAdapter = _

              New OleDb.OleDbDataAdapter("Select * From Contacts", m_cnADONetConnection)

        m_cbCommandBuilder = New OleDb.OleDbCommandBuilder(m_daDataAdapter)

        m_daDataAdapter.Fill(m_dtContacts)

        Me.ShowCurrentRecord()

    End Sub

    Private Sub btnMoveFirst_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnMoveFirst.Click

        ' Move to the first row and show the data.

        m_rowPosition = 0

        Me.ShowCurrentRecord()

    End Sub

    Private Sub btnMovePrevious_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnMovePrevious.Click

        ' If not at the first row, go back one row and show the record.

        If m_rowPosition > 0 Then

            m_rowPosition = m_rowPosition - 1

            Me.ShowCurrentRecord()

        End If

    End Sub

    Private Sub btnMoveNext_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnMoveNext.Click

        ' If not on the last row, advance one row and show the record.

        If m_rowPosition < (m_dtContacts.Rows.Count - 1) Then

            m_rowPosition = m_rowPosition + 1

            Me.ShowCurrentRecord()

        End If

    End Sub

    Private Sub btnMoveLast_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnMoveLast.Click

        ' If there are any rows in the data table, move to the last and show the record.

        If m_dtContacts.Rows.Count > 0 Then

            m_rowPosition = m_dtContacts.Rows.Count - 1

            Me.ShowCurrentRecord()

        End If

    End Sub

    Private Sub btnSave_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnSave.Click

        ' If there is existing data, update it.

        If m_dtContacts.Rows.Count <> 0 Then

            m_dtContacts.Rows(m_rowPosition)("ContactName") = txtContactName.Text

            m_dtContacts.Rows(m_rowPosition)("State") = txtState.Text

            m_daDataAdapter.Update(m_dtContacts)

        End If

    End Sub

    Private Sub btnAddNew_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnAddNew.Click

        Dim drNewRow As DataRow = m_dtContacts.NewRow()

        drNewRow("ContactName") = txtNewContactName.Text

        drNewRow("State") = txtNewState.Text

        m_dtContacts.Rows.Add(drNewRow)

        m_daDataAdapter.Update(m_dtContacts)

        m_rowPosition = m_dtContacts.Rows.Count - 1

        Me.ShowCurrentRecord()

    End Sub

    Private Sub btnDelete_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnDelete.Click

        ' If there is data, delete the current row.

        If m_dtContacts.Rows.Count <> 0 Then

            m_dtContacts.Rows(m_rowPosition).Delete()

            m_daDataAdapter.Update(m_dtContacts)

            m_rowPosition = 0

            Me.ShowCurrentRecord()

        End If

    End Sub

End Class

Simplified Summers ATM machine
Database:

Family Given SavingsAcct | _SavingsBal | ChequeAcct | ChequeBal

McLoughlan _ Trevor 12902 5650 12903 655
3061 Mahoud Yusef 12008 2995 12009 6750
4339 Calle Rhianna 12328 5790 12329 450
5114 Lee Melarie 12548 2480 12549 665
5213 Collins Robert 12873 1280 12674 3200
7601 OConner  Michelle 12861 2300 12662 5420

The Customer database consists of seven fields including PIN, the customer name and balances for Cheque and Savings accounts. Note, the account numbers are not used by the application.

Interface:

The VB.Net interface has been simplified. It consists only of a label for the screen (lblScreen), buttons (btn0, btn1, btn2, ……..btn9, btnClear, btnOK, btnCancel, btnExit) and two radio buttons (rdbCheque, rdbSavings) as shown. The number pad is enclosed in a Group Box.

Once your design elements have been created, enter the code given below. Hint! Copy and paste the code located inside the subroutines, not the subroutine “shell”.

VB.Net Code:

Imports System.Data.OleDb

Public Class Form1

    '-----------------------------------------------------------------------------

    ' Copyright for this module belongs to the author. You may freely use, copy      

    ' and distribute for educational purposes only but not for commercial purposes.  

    '

    ' Author: Harry Schlanger

    ' Date: 2/07/07

    ' Location: Belmont High School

    ' Email: big_koala@hotmail.com

    ' Purpose: Introductory database material for Unit 4, Software Development

    ' Description: This program is an Automatic Teller Machine (ATM) system to 

    '              withdraw money from either Cheque strAccount or Savings strAccount. 

    '              Money has to exist in the strAccount. The ATM expects a four digit 

    '              PIN to be entered by the user. If accepted, it will ask for $$ 

    '              to withdraw. The balance will then be returned to the screen.

    '

    '              The interface is linked to an Access databse called CreditUnion.mdb 

    '              located in the \bin folder of the VB.Net project. If moved from 

    '              this location, a path to this database will need to be specified.

    '

    '              The software is based on the text by Graeme Summers 2003 Third Ed.

    '              "Programming with Visual Basic. Introduction to VB.Net"

    '              P176. Except that we have simplified it as much as possible AND 

    '              proper naming convention has been introduced. We have also placed

    '              internal comments to explain the workings of the code.

    '

    '-----------------------------------------------------------------------------

    'Declarations at module level

    Dim adpCustomer As OleDbDataAdapter

    Dim conCreditUnion As OleDbConnection

    Dim datCustomer As New DataSet()

    Dim strPIN, strAccount As String, sngChequeBal, sngSavingsBal As Single

    Dim blnPINEntry As Boolean = True, blnFirstDigit As Boolean = True

    Dim strWithDrawAmt As String = ""

    Private Function blnValidPIN() As Boolean

        '-------------------------------------------------------------------------------

        ' The key to opening the application is for this function to be true!!

        ' The dataset containing Customer table is searched for the PIN and the row

        ' of data containing that PIN is returned and saved in rowCurrent

        ' (It's equivalent to running a query but everything is done automatically

        '  behind the scenes by the ADO and OleDb technologies)

        ' If the row is found - the application is unlocked. If no row is found,

        ' the application is not opened.

        ' The value of the function is examined where the function is called. That is,

        ' in btnOK_Click()

        '---------------------------------------------------------------------------------

        Dim rowCurrent As DataRow = datCustomer.Tables("tblCustomers").Rows.Find(strPIN)

        Dim blnFound As Boolean = Not (rowCurrent Is Nothing)

        ' if data exists in the table then continue on...

        ' Get the field data from the record and store them in the program variables

        If blnFound Then

            sngChequeBal = rowCurrent.Item("ChequeBal")

            sngSavingsBal = rowCurrent.Item("SavingsBal")

            If rdbSavings.Checked Then

                strAccount = rowCurrent.Item("SavingsAcct")

            Else

                strAccount = rowCurrent.Item("ChequeAcct")

            End If

            'Display the user's name to validate the PIN acceptance

     ' write your own code here

        End If

        Return blnFound

    End Function

    Private Sub WasteTime(ByVal N As Integer)

        ' The value of N is the number of seconds to waste

        Dim Start As Date

        Start = TimeOfDay

        Do

        Loop Until DateDiff(DateInterval.Second, Start, TimeOfDay) >= N

    End Sub

    Private Sub Form1_Load(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MyBase.Load

        '-------------------------------------------------------------------------------

        ' Connect to the database, then extract (or update) data from (to) the database

        ' via a data adapter named adpCustomer.

        ' The data is extracted (updated) using a SQL instruction located in the .CommandText 

        ' method of the data adapter's .SelectCommand method.

        ' The queried data is placed in the program's DataSet() named datCustomer

        ' for easy access in the program.

        ' Key(0) is an array with index 0 to store a column of data

        '------------------------------------------------------------------------------

        Dim Key(0) As DataColumn

        Dim ConnectString As String = "Provider=Microsoft.Jet.OLEDB.4.0;" & _

        "Data Source=CreditUnion.mdb"

        conCreditUnion = New OleDbConnection(ConnectString)

        conCreditUnion.Open()

        adpCustomer = New OleDbDataAdapter()

        adpCustomer.SelectCommand = New OleDbCommand()

        ' use the data adapter to run a query against the database

        With adpCustomer.SelectCommand

            .Connection = conCreditUnion

            .CommandText = "SELECT * FROM tblCustomers"

            .CommandType = CommandType.Text

            .ExecuteNonQuery()

        End With

        ' place the data into the DataSet named datCustomer for easy access

        ' then return only data that is in the first column (with index 0)

        adpCustomer.Fill(datCustomer, "tblCustomers")

        Key(0) = datCustomer.Tables("tblCustomers").Columns(0)

        datCustomer.Tables("tblCustomers").PrimaryKey = Key

    End Sub

    Private Sub UpdateBalance()

        ' Here we have to interact with the database as we wish to update data using an

        ' UPDATE query. To test if this works, it's best to first create the query in Access,

        ' then get the SQL from Access and copy it here.

        '

        adpCustomer.SelectCommand = New OleDbCommand()

        With adpCustomer.SelectCommand

            .Connection = conCreditUnion

            .CommandText = "UPDATE tblCustomers SET SavingsBal = " + sngSavingsBal.ToString + ", ChequeBal = " + sngChequeBal.ToString + " WHERE PIN = '" + strPIN + "'"

            .CommandType = CommandType.Text

            .ExecuteNonQuery()

        End With

        ' Refresh the data - use the data adapter to place the new tblCustomers data into 

        ' the dataset datCustomer

        adpCustomer.Fill(datCustomer, "tblCustomers")

    End Sub

    Private Sub Transaction()

        Dim blnTooLow As Boolean

        blnTooLow = (rdbSavings.Checked And Val(strWithDrawAmt) > sngSavingsBal) Or (rdbCheque.Checked And Val(strWithDrawAmt) > sngChequeBal)

        With lblScreen

            If blnTooLow Then  ' not enough in strAccount

                .Text = "Balance too low"

                .Refresh()

                WasteTime(2)

            Else    ' enough to cover withdrawal

                .Text = "Transaction in progress"

                .Refresh()

                WasteTime(4)

                ' reduce amount in chosen strAccount

                If rdbSavings.Checked Then

                    sngSavingsBal -= Val(strWithDrawAmt)




' enter your own code here




' display the Savings balance to the screen

                Else

                    sngChequeBal -= Val(strWithDrawAmt)



' enter your own code here




' display the Cheque balance to the screen

                End If

                UpdateBalance()

            End If

            blnPINEntry = True

            blnFirstDigit = True

            strWithDrawAmt = ""

            strPIN = ""

            .Text = "Enter your PIN"

        End With

    End Sub

    Private Sub btnOK_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnOK.Click

        Static intBadPINCount As Integer

        blnFirstDigit = True

        ' If the digits are entered and valid, a transaction occurs with the database

        ' However, if there is a problem with the PIN or if there are too many attempts

        ' at entering the PIN, then a message is returned to the user via the Screen.

        With lblScreen

            If blnPINEntry Then    ' entering PIN

                If blnValidPIN() Then

                    blnPINEntry = False

                    .Text = "Enter amount in $"

                Else    ' invalid PIN entry

                    strPIN = ""

                    intBadPINCount += 1

                    .Text = "Invalid PIN"

                    .Refresh()

                    WasteTime(2)

                    If intBadPINCount = 4 Then

                        intBadPINCount = 0

                        .Text = "Card taken, see manager"

                        .Refresh()

                        WasteTime(2)

                    End If

                    .Text = "Enter your PIN"

                End If

            Else    ' withdrawal amount entry (only multiples of $5)

                If Val(strWithDrawAmt) Mod 5 > 0 Then

                    .Text = "Invalid amount"

                    .Refresh()

                    WasteTime(2)

                    strWithDrawAmt = ""

                    .Text = "Enter amount in multiples of $5"

                Else    ' is a multiple of $5

                    ' modify the data in the database

                    Transaction()

                End If

            End If

        End With

    End Sub

    Private Sub btnClear_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnClear.Click

        blnFirstDigit = True

        If blnPINEntry Then    ' entering PIN

            lblScreen.Text = "Enter your PIN"

            strPIN = ""

        Else    ' entering amount to withdraw

            lblScreen.Text = "Enter amount in $"

            strWithDrawAmt = ""

        End If

    End Sub

    Private Sub btnExit_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnExit.Click

        conCreditUnion.Close()

        End

    End Sub

    Private Sub btn1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btn1.Click, btn0.Click, _

    btn2.Click, btn3.Click, btn4.Click, btn5.Click, _

    btn6.Click, btn7.Click, btn8.Click, btn9.Click

        ' At the start of the program both blnFirstDigit and blnPINEntry are set to TRUE

        ' The entered number is stored into strPIN using the system object sender.Text

        ' and an * is returned to the Screen. blnFirstDigit is set to FALSE so that next

        ' digit that is read does not wipe the Screen but instead shows ** and the digit

        ' accumulates into strPIN

        ' ...and so on

        If blnFirstDigit Then  ' starting entry

            lblScreen.Text = ""

        End If

        If blnPINEntry Then

            lblScreen.Text += "*"

            strPIN += sender.Text

        Else    ' entering amount to withdraw

            lblScreen.Text += sender.Text

            strWithDrawAmt += sender.Text

        End If

        blnFirstDigit = False

    End Sub

    Private Sub btnCancel_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnCancel.Click

' add your own code here
    End Sub

End Class
Questions and tasks:

Find the spot in the code where you should add your own code to answer these questions

1. What occurs in these lines of code: 
conCreditUnion = New OleDbConnection(ConnectString)
conCreditUnion.Open()
Ans: 
2. In this set of instructions, the .CommandText instruction contains a query that will act upon the database. What data do you expect to be returned to the data adapter?

        With adpCustomer.SelectCommand

            .Connection = conCreditUnion

            .CommandText = "SELECT * FROM tblCustomers"

            .CommandType = CommandType.Text

            .ExecuteNonQuery()

        End With

Ans: 

3. Add your own code to the Cancel button. Hint! you have to reset the Boolean flag that controls the PIN entry and have a suitable screen message

Ans:

4. Display the Savings or Cheque balance to the screen after a transaction is made. Hint! Go to the Transaction subroutine to add your code there.
Ans:
5. Display the user's name after validation of the PIN acceptance. Hint! Write your code in the function blnValidPIN().
Ans:
